

Normalization by evaluation for $\lambda^{\rightarrow 2}$

Thorsten Altenkirch¹ * and Tarmo Uustalu² **

¹ School of Computer Science and IT, University of Nottingham
Nottingham NG8 1BB, United Kingdom
`txa@cs.nott.ac.uk`

² Institute of Cybernetics, Tallinn Technical University
Akadeemia tee 21, EE-12618 Tallinn, Estonia
`tarmo@cs.ioc.ee`

Abstract. We show that the set-theoretic semantics for $\lambda^{\rightarrow 2}$ is complete by inverting evaluation using decision trees. This leads to an implementation of normalization by evaluation which is witnessed by the source of part of this paper being a literate Haskell script. We show the correctness of our implementation using logical relations.

1 Introduction

Which is the simplest typed λ -calculus without uninterpreted base types or type variables? We suggest that the answer should be $\lambda^{\rightarrow 2}$: simply typed lambda calculus extended by the type of booleans `Bool` with `True, False : Bool` and `If t u0 u1 : σ` , given `t : Bool` and `u0, u1 : σ` . The equational theory is given by the usual $\beta\eta$ -equations of λ^{\rightarrow} and the following equations concerning `Bool`:

$$\begin{aligned} \text{If True } u_0 u_1 &=_{\beta\eta} u_0 \\ \text{If False } u_0 u_1 &=_{\beta\eta} u_1 \\ \text{If } t \text{ True False} &=_{\beta\eta} t \\ v (\text{If } t u_0 u_1) &=_{\beta\eta} \text{If } t (v u_0) (v u_1) \end{aligned}$$

The equations are motivated by the categorical interpretation of `Bool` as a boolean object, i.e., an object `Bool` such that $\text{Hom}(\Gamma \times \text{Bool}, A) \simeq \text{Hom}(\Gamma, A) \times \text{Hom}(\Gamma, A)$ (naturally in Γ and A). The calculus can thus be interpreted in any cartesian closed category with `Bool` (using the cartesian structure to interpret contexts).

The equational theory introduces some interesting equalities. E.g., consider

$$\begin{aligned} \text{once} &= \lambda f : \text{Bool} \rightarrow \text{Bool}. \lambda x : \text{Bool}. f x \\ \text{thrice} &= \lambda f : \text{Bool} \rightarrow \text{Bool}. \lambda x : \text{Bool}. f (f (f x)) \end{aligned}$$

* partially supported by EU Framework 5 thematic networks TYPES and APPSEM II and the Estonian IT Foundation

** partially supported by the Estonian Science Foundation under grant No. 5567

We observe that `once` $=_{\beta\eta}$ `thrice`. To see this, we note that, given $f : \text{Bool} \rightarrow \text{Bool}$, we have

$$\begin{aligned}
f(f(f \text{True})) &=_{\beta\eta} \text{If } (f \text{True}) (f(f \text{True})) (f(f \text{False})) \\
&=_{\beta\eta} \text{If } (f \text{True}) \text{True } (f(f \text{False})) \\
&=_{\beta\eta} \text{If } (f \text{True}) \text{True } (\text{If } (f \text{False}) (f \text{True}) (f \text{False})) \\
&=_{\beta\eta} \text{If } (f \text{True}) \text{True } (\text{If } (f \text{False}) \text{False False}) \\
&=_{\beta\eta} \text{If } (f \text{True}) \text{True False} \\
&=_{\beta\eta} f \text{True}
\end{aligned}$$

Symmetrically, we can show that $f(f(f \text{False})) =_{\beta\eta} f \text{False}$, and hence

$$\begin{aligned}
&\text{thrice} \\
&= \lambda f : \text{Bool} \rightarrow \text{Bool}. \lambda x : \text{Bool}. f(f(f x)) \\
&=_{\beta\eta} \lambda f : \text{Bool} \rightarrow \text{Bool}. \lambda x : \text{Bool}. \text{If } x (f(f(f \text{True}))) (f(f(f \text{False}))) \\
&=_{\beta\eta} \lambda f : \text{Bool} \rightarrow \text{Bool}. \lambda x : \text{Bool}. \text{If } x (f \text{True}) (f \text{False}) \\
&=_{\beta\eta} \lambda f : \text{Bool} \rightarrow \text{Bool}. \lambda x : \text{Bool}. f x \\
&= \text{once}
\end{aligned}$$

It is easy to see that `once` and `thrice` are equal in the standard semantics where `Bool` is interpreted by a two-element set $\text{Bool} = \{\text{true}, \text{false}\}$ and function types are set-theoretic function spaces. We observe that there are only four elements in $\text{Bool} \rightarrow \text{Bool} = \{x \mapsto \text{true}, x \mapsto x, x \mapsto \neg x, x \mapsto \text{false}\}$ and that for all the four $f \in \text{Bool} \rightarrow \text{Bool}$ we have $f^3 = f$.

May we use set-theoretic reasoning to prove equalities up to $\beta\eta$ -convertibility? The answer is yes for λ^{-2} , because for λ^{-2} we can *invert* set-theoretic *evaluation* of typed closed terms. That is: we can define a function $\text{quote}^\sigma \in \llbracket \sigma \rrbracket_{\text{set}} \rightarrow \text{Tm } \sigma$ such that $t =_{\beta\eta} \text{quote}^\sigma \llbracket t \rrbracket_{\text{set}}$, for any $t \in \text{Tm } \sigma$. Consequently, we get that, for any $t, t' \in \text{Tm } \sigma$, $t =_{\beta\eta} t' \iff \llbracket t \rrbracket_{\text{set}} = \llbracket t' \rrbracket_{\text{set}}$.

The existence of `quote` also implies that $=_{\beta\eta}$ is maximally consistent, i.e., identifying any two non- $\beta\eta$ -convertible closed terms would lead to an inconsistent theory. This provides another justification for the specific choice of $=_{\beta\eta}$.

We do not analyze the normal forms here, i.e. the codomain of `nf` and `quote` here. However, the construction presented here, which is based on decision trees, leads to simple normal forms and we conjecture that this is the same set as the set of normal forms presented in [1, 5] in the case $\text{Bool} = 1 + 1$.

Haskell as a poor man's Type Theory

Our construction is entirely constructive, so it can be carried out, e.g., in Martin-Löf's Type Theory, and we obtain an implementation of normalization $\text{nf}^\sigma t = \text{quote}^\sigma \llbracket t \rrbracket_{\text{set}}$. We shall here use the functional language Haskell as a poor man's Type Theory and obtain a Haskell program to normalize terms.

Haskell hasn't got dependent types (in particular inductive families), hence the Haskell types we are using are only approximations of their type-theoretic

correspondents. E.g., the type-theory type $\mathsf{Tm}_\Gamma \sigma$ contains all welltyped terms of type σ in context Γ , but its Haskell counterpart Tm contains all untyped terms. Similarly, the set-theoretic denotation of a type $\sigma \rightarrow \tau$ is given by $\llbracket \sigma \rightarrow \tau \rrbracket_{\text{set}} = \llbracket \sigma \rrbracket_{\text{set}} \rightarrow \llbracket \tau \rrbracket_{\text{set}}$ but its Haskell implementation is given by a recursive type $\mathsf{E1}$ with a constructor $\mathsf{SLam} \in \mathsf{Ty} \rightarrow (\mathsf{E1} \rightarrow \mathsf{E1}) \rightarrow \mathsf{E1}$.

We believe that this informal use of Type Theory is an effective way to arrive at functional programs which are correct by construction. However, we hope that in the future we can go further and bridge the gap between informal type theoretic reasoning and the actual implementation by using a dependently typed programming language. such as the Epigram system, which is currently being developed by Conor McBride [12].

Related work

Inverting evaluation to achieve normalization by evaluation (NBE, aka. reduction-free normalization) was pioneered in [6] for simply typed lambda calculus with type variables and a non-standard semantics; a categorical account in terms of presheaves was given in [2]; this was extended to System F in [3, 4]; see [9] for a recent survey on NBE. The completeness of the set-theoretic model in the presence of coproducts has been shown in [8] and our case arises as a special case when there are no type variables. Normalization procedures for typed λ -calculus with coproducts can be found in [10, 11] using rewriting techniques and [1, 5] using NBE and sheaf theory. Both approaches allow type variables but do not handle the empty type. Here we present a much simpler construction for closed types using the simplest possible semantics of first-order simply typed λ -calculi—the set-theoretic one—and also provide a concrete implementation of `quote` and `nf`, whose correctness we show in detail.

2 Implementation of $\lambda \rightarrow^2$

The source of Sections 2 and 3 of this paper is a literate Haskell script implementing normalization for $\lambda \rightarrow^2$ and is available from

<http://www.cs.nott.ac.uk/~txa/publ/Nbe2.lhs>

We start by introducing types $\mathsf{Ty} \in \star$, variables $\mathsf{Var} \in \star$, typing contexts $\mathsf{Con} \in \star$ and untyped terms $\mathsf{Tm} \in \star$ of the object language by the following Haskell datatype definitions:

```
data Ty = Bool | Ty :-> Ty
      deriving (Show, Eq)
```

```
type Var = String
```

```
type Con = [ (Var, Ty) ]
```

```

data Tm = Var Var
        | TTrue | TFalse | If Tm Tm Tm
        | Lam Ty String Tm | App Tm Tm
        deriving (Show, Eq)

```

We view these recursive definitions as inductive definitions, i.e., we do not consider infinite terms. All the functions we define are total wrt. their precise type-theoretic types.

Implementing typed terms $Tm \in Con \rightarrow Ty \rightarrow \star$ would take inductive families, which we cannot use in Haskell. But we can implement type inference $infer \in Con \rightarrow Tm \rightarrow Maybe Ty$ (where $Maybe X = 1 + X$ as usual):

```

infer :: Con -> Tm -> Maybe Ty
infer gamma (Var x) =
    do sigma <- lookup x gamma
       Just sigma
infer gamma TTrue  = Just Bool
infer gamma TFalse = Just Bool
infer gamma (If t u0 u1) =
    do Bool <- infer gamma t
       sigma0 <- infer gamma u0
       sigma1 <- infer gamma u1
       if sigma0 == sigma1 then Just sigma0 else Nothing
infer gamma (Lam sigma x t) =
    do tau <- infer ((x, sigma) : gamma) t
       Just (sigma :-> tau)
infer gamma (App t u) =
    do (sigma :-> tau) <- infer gamma t
       sigma' <- infer gamma u
       if sigma == sigma' then Just tau else Nothing

```

This implementation is correct in the sense that $t \in Tm_{\Gamma} \sigma$ iff $infer_{\Gamma} t = Just \sigma$.

Evaluation of types $\llbracket - \rrbracket \in Ty \rightarrow \star$ is again an inductive family, which we cannot implement in Haskell, and the workaround is to have all $\llbracket \sigma \rrbracket$ coalesced into one metalanguage type el (of untyped elements) much the same way as all $Tm_{\Gamma} \sigma$ appear coalesced in Tm . We use a type class `Sem` to state what we require of such a coalesced type el :

```

class Sem el where
    true  :: el
    false :: el
    xif :: el -> el -> el -> el
    lam :: Ty -> (el -> el) -> el
    app :: el -> el -> el

```

Evaluation of types $\llbracket - \rrbracket \in Ty \rightarrow \star$ naturally induces evaluation of contexts $\llbracket - \rrbracket \in Con \rightarrow \star$, defined by

$$\frac{}{\llbracket \cdot \rrbracket \in \llbracket \llbracket \cdot \rrbracket \rrbracket} \quad \frac{\rho \in \llbracket \Gamma \rrbracket \quad d \in \llbracket \sigma \rrbracket}{(x, d) : \rho \in \llbracket (x, \sigma) : \Gamma \rrbracket}$$

We write $\text{Tm } \Gamma = \llbracket \Gamma \rrbracket_{\text{syn}}$ for the set of closed substitutions, which arises as in instance when using $\llbracket \sigma \rrbracket_{\text{syn}} = \text{Tm}_{\llbracket \cdot \rrbracket} \sigma$ as the interpretation of types.

In the Haskell code we approximate evaluation of contexts by `Env`:

```
type Env e1 = [ (Var, e1) ]
```

Given $t \in \text{Tm}_{\Gamma} \sigma$ we define the evaluation of terms $\llbracket t \rrbracket \in \llbracket \Gamma \rrbracket \rightarrow \llbracket \sigma \rrbracket$. In Haskell this is implemented as `eval`:

```
eval :: Sem e1 => Env e1 -> Tm -> e1
eval rho (Var x) = d
  where (Just d) = lookup x rho
eval rho TTrue = true
eval rho TFalse = false
eval rho (If t u0 u1) =
  xif (eval rho t) (eval rho u0) (eval rho u1)
eval rho (Lam sigma x t) =
  lam sigma (\ d -> eval ((x, d) : rho) t)
eval rho (App t u) = app (eval rho t) (eval rho u)
```

The standard set-theoretic semantics is given by

$$\begin{aligned} \llbracket \text{Bool} \rrbracket_{\text{set}} &= \text{Bool} \\ \llbracket \sigma : \rightarrow \tau \rrbracket_{\text{set}} &= \llbracket \sigma \rrbracket_{\text{set}} \rightarrow \llbracket \tau \rrbracket_{\text{set}} \end{aligned}$$

This can be represented in Haskell as an instance of `Sem`:

```
data El = STrue | SFalse | SLam Ty (El -> El)
```

```
instance Sem El where
  true  = STrue
  false = SFalse
  xif STrue d _ = d
  xif SFalse _ d = d
  lam = SLam
  app (SLam _ f) d = f d
```

Since sets form a cartesian closed category with a boolean object, the set-theoretic semantics validates all $\beta\eta$ -equalities. This is to say that $\llbracket - \rrbracket_{\text{set}}$ is equationally sound:

Proposition 2.1 (Soundness). *If $\rho \in \llbracket \Gamma \rrbracket$ and $t =_{\beta\eta} t' \in \text{Tm}_{\Gamma} \sigma$, then $\llbracket t \rrbracket_{\text{set}} \rho = \llbracket t' \rrbracket_{\text{set}} \rho$.*

Since all the sets we consider are finite, semantic equality can be implemented in Haskell, by making use of the function `enum` $\in (\sigma \in \text{Ty}) \rightarrow \text{Tree } \llbracket \sigma \rrbracket$, which we will provide later:

```

instance Eq El where
  STrue  == STrue  = True
  SFalse == SFalse = True
  (SLam sigma f) == (SLam _ f') =
    and [f d == f' d | d <- flatten (enum sigma)]
  _ == _ = False

```

Using on the same function we can also print elements of El:

```

instance Show El where
  show STrue  = "STrue"
  show SFalse = "SFalse"
  show (SLam sigma f) =
    "SLam " ++ (show sigma) ++ " " ++
    (show [ (d, f d) | d <- flatten (enum sigma) ])

```

The equational theory of the calculus itself gives rise to another semantics—the free semantics, or typed terms up to $\beta\eta$ -convertibility. This can be approximated by the following Haskell code, which uses a redundancy-avoiding version `if'` of `If` which produces a shorter but $\beta\eta$ -equivalent term:

```

if' :: Tm -> Tm -> Tm -> Tm
if' t TTrue TFalse = t
if' t u0 u1 = if u0 == u1 then u0 else If t u0 u1

```

```

instance Sem Tm where
  true  = TTrue
  false = TFalse
  xif = if'
  lam sigma f = Lam sigma "x" (f (Var "x"))
  app = App

```

We also observe that the use of a fixed variable is only justified by the fact that our algorithm uses at most one bound variable at the time. A correct dependently typed version of the free semantics requires the use of presheaves to ensure that the argument to `Lam` is stable under renaming. We refrain from presenting the details here. It is well known that this semantics is equationally sound.

3 Implementation of quote

We now proceed to implementing $\text{quote} \in (\sigma \in \text{Ty}) \rightarrow \llbracket \sigma \rrbracket_{\text{set}} \rightarrow \text{Tm } \sigma$.

To define $\text{quote}^{\sigma \rightarrow \tau}$ we use enum^{σ} , which generates a decision tree whose leaves are all the elements of $\llbracket \sigma \rrbracket$, and $\text{questions}^{\sigma}$, which generates the list of questions, i.e. elements of $\llbracket \sigma \rrbracket \rightarrow \llbracket \text{Bool} \rrbracket$, based on answers to whom an element of $\llbracket \sigma \rrbracket$ can be looked up in the tree enum^{σ} . (Since our decision trees are perfectly balanced and we use the same list questions along each branch of a tree, we separate the questions labelling from the tree.)

Decision trees $\text{Tree} \in \text{Ty} \rightarrow \star$ are provided by

```
data Tree a = Val a | Choice (Tree a) (Tree a) deriving (Show, Eq)
```

We will exploit the fact that Tree is a monad

```
instance Monad Tree where
  return = Val
  (Val a) >>= h = h a
  (Choice l r) >>= h = Choice (l >>= h) (r >>= h)
```

(return and >>= are Haskell for the unit resp. the bind or Kleisli extension operation of a monad) and hence a functor

```
instance Functor Tree where
  fmap h ds = ds >>= return . h
```

(fmap is Haskell for the action of a functor on morphisms).

It is convenient to use the function flatten which calculates the list of leaves of a given tree:

```
flatten :: Tree a -> [ a ]
flatten (Val a) = [ a ]
flatten (Choice l r) = (flatten l) ++ (flatten r)
```

We implement enum^σ and questions^σ by mutual induction on $\sigma \in \text{Ty}$. The precise typings of the functions are $\text{enum} \in (\sigma \in \text{Ty}) \rightarrow \text{Tree } \llbracket \sigma \rrbracket$ and $\text{questions} \in (\sigma \in \text{Ty}) \rightarrow \llbracket \sigma \rrbracket \rightarrow \llbracket \text{Bool} \rrbracket$. As usual, Haskell cannot express those subtleties due to its lack of dependent types, but we can declare

```
enum :: Sem e1 => Ty -> Tree e1
questions :: Sem e1 => Ty -> [ e1 -> e1 ]
```

The base case is straightforward: A boolean is true or false and to know which one it is it suffices to know it.

```
enum Bool = Choice (Val true) (Val false)
```

```
questions Bool = [ \ b -> b ]
```

The implementation of $\text{enum}^{\sigma \rightarrow \tau}$ and $\text{questions}^{\sigma \rightarrow \tau}$ proceeds from the idea that a function is determined by its graph: to know a function it suffices to know its value on all possible argument values. The main idea in the implementation of $\text{enum}^{\sigma \rightarrow \tau}$ is therefore to start with enum^τ and to duplicate the tree for each question in questions^σ using the bind of Tree:

```
enum (sigma :-> tau) =
  fmap (lam sigma) (mkEnum (questions sigma) (enum tau))
```

```
mkEnum :: Sem e1 => [ e1 -> e1 ] -> Tree e1 -> Tree (e1 -> e1)
```

```
mkEnum [] es = fmap (\ e -> \ d -> e) es
```

```
mkEnum (q : qs) es = (mkEnum qs es) >>= \ f1 ->
  (mkEnum qs es) >>= \ f2 ->
  return (\ d -> xif (q d) (f1 d) (f2 d))
```

$\text{questions}^{\sigma \rightarrow \tau}$ produces the appropriate questions by enumerating σ and using questions from τ :

```
questions (sigma :-> tau) =
  [ \ f -> q (app f d) | d <- flatten (enum sigma),
    q <- questions tau ]
```

As an example, the enumeration and questions for $\text{Bool} \rightarrow \text{Bool}$ return:

```
Choice
  (Choice
    (Val (lam Bool (\ d -> xif d true true)))
    (Val (lam Bool (\ d -> xif d true false))))
  (Choice
    (Val (lam Bool (\ d -> xif d false true )))
    (Val (lam Bool (\ d -> xif d false false))))
```

resp.

```
(\ f -> app f true :
  (\ f -> app f false :
    []))
```

We can look up an element in the decision tree for a type by answering all the questions, this is realized by the function `find` below. To define the domain of `find` precisely we define a relation between lists of questions and trees of answers $\diamond \subseteq [a] \times \text{Tree } b$ inductively:

$$\frac{}{[] \diamond (\text{Val } t)} \quad \frac{as \diamond l \quad as \diamond r}{a : as \diamond \text{Choice } l \ r}$$

Now given $as \in [[\text{Bool}]]$, $ts \in \text{Tree } [\sigma]$, s.t. $as \diamond ts$ we obtain $\text{find } as \ ts \in [\sigma]$, implemented in Haskell:

```
find :: Sem e1 => [ e1 ] -> Tree e1 -> e1
find [] (Val t) = t
find (a : as) (Choice l r) = xif a (find as l) (find as r)
```

We are now ready to implement $\text{quote}^{\sigma} \in [[\sigma]_{\text{set}}] \rightarrow \text{Tm } \sigma$, with Haskell typing

```
quote :: Ty -> El -> Tm
```

by induction on $\sigma \in \text{Ty}$. As usual, the base case is easy:

```
quote Bool STrue  = TTrue
quote Bool SFalse = TFalse
```

$\text{quote}^{\sigma \rightarrow \tau}$ is more interesting: Our strategy is to map $\text{quote}^{\tau} \circ f$ to the set-theoretic enum^{τ} and to then build a tree of `If` expressions by using the syntactic $\text{questions}^{\sigma}$ in conjunction with the syntactic `find`:

```

quote (sigma :-> tau) (SLam _ f) =
  lam sigma (\ t -> find [ q t | q <- questions sigma ]
                      (fmap (quote tau . f) (enum sigma)))

```

(Notice that in Haskell it is inferred automatically which semantics is meant where.)

As already discussed in the introduction, we implement normalization $\text{nf} \in (\sigma \in \text{Ty}) \rightarrow \text{Tm } \sigma \rightarrow \text{Tm } \sigma$ by

```

nf :: Ty -> Tm -> Tm
nf sigma t = quote sigma (eval [] t)

```

Since we can infer types, we can implement $\text{nf}' \in \text{Tm} \rightarrow \text{Maybe } (\sum_{\sigma \in \text{Ty}} \text{Tm } \sigma)$:

```

nf' :: Tm -> Maybe (Ty, Tm)
nf' t = do sigma <- infer [] t
        Just (sigma, nf sigma t)

```

We test our implementation with the example from the introduction:

```

b2b = Bool :-> Bool
once  = Lam b2b "f" (Lam Bool "x" (App (Var "f") (Var "x")))
twice = Lam b2b "f" (Lam Bool "x" (App (Var "f")
                                       (App (Var "f") (Var "x"))))
thrice = Lam b2b "f"
        (Lam Bool "x" (App (Var "f")
                          (App (Var "f")
                                (App (Var "f") (Var "x"))))))

```

and convince ourselves that $(\text{nf}' \text{ once} = \text{nf}' \text{ thrice}) = \text{true}$ but $(\text{nf}' \text{ once} = \text{nf}' \text{ twice}) = \text{false}$. Since semantic equality is decidable we do not actually have to construct the normal forms to decide convertibility.

Since testing can only reveal the presence of errors we shall use the rest of this paper to prove that `quote` and hence `nf` behave correctly.

4 Correctness of quote

The main tool in our proof will be a notion of logical relations, a standard tool for the characterization of definable elements in models of typed lambda calculi since the pioneering work of Plotkin [13].

Let us agree to abbreviate $\text{Tm}_{\square} \sigma$ by $\text{Tm } \sigma$ and $\llbracket t \rrbracket_{\text{set}}$ by $\llbracket t \rrbracket$.

Definition 4.1 (Logical Relations). *We define a family of relations $R^{\sigma} \subseteq \text{Tm } \sigma \times \llbracket \sigma \rrbracket_{\text{set}}$ by induction on $\sigma \in \text{Ty}$ as follows:*

- $t R^{\text{Bool}} b$ iff $t =_{\beta\eta} \text{True}$ and $b = \text{true}$ or $t =_{\beta\eta} \text{False}$ and $b = \text{false}$;
- $t R^{\sigma \rightarrow \tau} f$ iff, for all u, d , $u R^{\sigma} d$ implies $\text{App } t \ u R^{\tau} f \ d$.

Note that R is not indexed by contexts, logical relations only relate closed terms.

We extend logical relations to contexts: Given $\Gamma \in \mathbf{Con}$ we define $R^\Gamma \subseteq \mathbf{Tm}\Gamma \times \llbracket \Gamma \rrbracket$ by:

$$\overline{\llbracket R^\Gamma \rrbracket} \quad \frac{\rho R^\Gamma \rho' \quad d R^\sigma d'}{(x, d) : \rho \quad R^{(x, \sigma) : \Gamma} \quad (x, d') : \rho'}$$

Logical relations are invariant under $\beta\eta$ -equality.

Lemma 4.2. *If $t R^\sigma d$ and $t =_{\beta\eta} t'$, then $t' R^\sigma d$.*

Logical relations obey the following Fundamental Theorem, a kind of soundness theorem for logical relations.

Lemma 4.3 (Fundamental Theorem of Logical Relations). *If $\theta R^\Gamma \rho$ and $t \in \mathbf{Tm}_\Gamma \sigma$, then $\llbracket t \rrbracket_{\text{set}} \theta R^\sigma \llbracket \rho \rrbracket_{\text{set}}$. In particular, if $t \in \mathbf{Tm} \sigma$, then $t R^\sigma \llbracket t \rrbracket_{\text{set}}$.*

The main result required to see that `quote` is correct is the following lemma:

Lemma 4.4 (Main Lemma). *If $t R^\sigma d$, then $t =_{\beta\eta} \text{quote}^\sigma d$.*

The proof of this lemma is the subject of the next section.

By correctness of `quote` we mean that it inverts set-theoretic evaluation of typed closed terms.

Theorem 4.5 (Main Theorem). *If $t \in \mathbf{Tm} \sigma$, then $t =_{\beta\eta} \text{quote}^\sigma \llbracket t \rrbracket_{\text{set}}$.*

Proof. Immediate from the Fundamental Theorem and the Main Lemma. \square

The (constructive) existence and correctness of `quote` has a number of straightforward important consequences.

Corollary 4.6 (Completeness). *If $t, t' \in \mathbf{Tm} \sigma$, then $\llbracket t \rrbracket_{\text{set}} = \llbracket t' \rrbracket_{\text{set}}$ implies $t =_{\beta\eta} t'$.*

Proof. Immediate from the Main Theorem. \square

From soundness (Proposition 2.1) and completeness together we get that $=_{\beta\eta}$ is decidable: checking whether $t =_{\beta\eta} t'$ reduces to checking whether $\llbracket t \rrbracket_{\text{set}} = \llbracket t' \rrbracket_{\text{set}}$, which is decidable as $\llbracket - \rrbracket_{\text{set}}$ is computable and equality in finite sets is decidable.

Corollary 4.7. *If $t, t' \in \mathbf{Tm} \sigma$, then $t =_{\beta\eta} t'$ iff $\text{quote}^\sigma \llbracket t \rrbracket_{\text{set}} = \text{quote}^\sigma \llbracket t' \rrbracket_{\text{set}}$.*

Proof. Immediate from soundness (Proposition 2.1) and the Main Theorem. \square

This corollary shows that $\text{nf}^\sigma = \text{quote}^\sigma \circ \llbracket - \rrbracket_{\text{set}} : \mathbf{Tm} \sigma \rightarrow \mathbf{Tm} \sigma$ indeed makes sense as normalization function: apart from just delivering, for any given typed closed term, some $\beta\eta$ -equal term, it is actually guaranteed to deliver the same term for t, t' , if t, t' are $\beta\eta$ -equal (morally, this is Church-Rosser for reduction-free normalization).

Note that although we only stated completeness and normalization for typed closed terms above, these trivially extend to all typed terms as opens terms can always be closed up by lambda-abstractions and this preserves $\beta\eta$ -equality.

Corollary 4.8. *If $t, t' \in \text{Tm } \sigma$ and $[C] [(x, t)] =_{\beta\eta} [C] [(x, t')]$ for every $C : \text{Tm}_{[(x, \sigma)]} \text{Bool}$, then $t =_{\beta\eta} t'$. Or, contrapositively, and more concretely, if $t, t' \in \text{Tm } (\sigma_1 \rightarrow \dots \rightarrow \sigma_n \rightarrow \text{Bool})$ and $t \neq_{\beta\eta} t'$, then there exist $u_1 \in \text{Tm } \sigma_1, \dots, u_n \in \text{Tm } \sigma_n$ such that*

$$\text{nf}^{\text{Bool}} (\text{App } (\dots (\text{App } t u_1) \dots) u_n) \neq \text{nf}^{\text{Bool}} (\text{App } (\dots (\text{App } t' u_1) \dots) u_n)$$

Proof. This corollary does not follow from the statement of the Main Theorem, but it follows from its proof. \square

Corollary 4.9 (Maximal consistency). *If $t, t' \in \text{Tm } \sigma$ and $t \neq_{\beta\eta} t'$, then from the equation $t = t'$ as an additional axiom one would derive $\text{True} = \text{False}$.*

Proof. Immediate from the previous corollary. \square

5 Proof of the main lemma

We now present the proof of the main lemma which was postponed in the previous section. To keep the proof readable, we write enum_{set} , $\text{questions}_{\text{set}}$, find_{set} to emphasize the uses of the set-theoretic semantics instances of enum , questions , find , while the free semantics instances will be written as enum_{syn} , $\text{questions}_{\text{syn}}$, find_{syn} . We use the fact that any functor F such as Tree has an effect on relations $R \subseteq A \times B$ denoted by $FR \subseteq FA \times FB$, which can be defined as:

$$\frac{z \in F\{(a, b) \in A \times B \mid aRb\}}{\text{fmap fst } z \text{ FR fmapsnd } z}$$

We first give the core of the proof and prove the lemmas this takes afterwards.

Proof (of the Main Lemma). By induction on σ .

- Case Bool : Assume $tR^{\text{Bool}}b$. Then either $t =_{\beta\eta} \text{True}$ and $b = \text{true}$, in which case we have

$$t =_{\beta\eta} \text{True} = \text{quote}^{\text{Bool}} \text{true} = \text{quote}^{\text{Bool}} b$$

or $t =_{\beta\eta} \text{False}$ and $b = \text{false}$, in which case we have

$$t =_{\beta\eta} \text{False} = \text{quote}^{\text{Bool}} \text{false} = \text{quote}^{\text{Bool}} b$$

- Case $\sigma \rightarrow \tau$: Assume $tR^{\sigma \rightarrow \tau}f$, for all u, d , $uR^\sigma d$ implies $\text{App } t uR^\tau f d$. We have

$$\begin{aligned} & t =_{\beta\eta} \text{Lam}^\sigma \text{ x } (\text{App } t (\text{Var } \text{x})) \\ & =_{\beta\eta} \text{ (by Lemma 5.2 below)} \\ & \quad \text{Lam}^\sigma \text{ x } (\text{App } t (\text{find}_{\text{syn}} [q (\text{Var } \text{x}) \mid q \leftarrow \text{questions}_{\text{syn}}^\sigma] \text{enum}_{\text{syn}}^\sigma)) \\ & = \text{Lam}^\sigma \text{ x } (\text{find}_{\text{syn}} [q (\text{Var } \text{x}) \mid q \leftarrow \text{questions}_{\text{syn}}^\sigma] \\ & \quad (\text{fmap } (\text{App } t) \text{enum}_{\text{syn}}^\sigma)) \\ & =_{\beta\eta} \text{ (by Sublemma)} \\ & \quad \text{Lam}^\sigma \text{ x } (\text{find}_{\text{syn}} [q (\text{Var } \text{x}) \mid q \leftarrow \text{questions}_{\text{syn}}^\sigma] \\ & \quad (\text{fmap } (\text{quote}^\tau \circ f) \text{enum}_{\text{set}}^\sigma)) \\ & = \text{quote}^{\sigma \rightarrow \tau} f \end{aligned}$$

The Sublemma is:

$$\text{fmap} (\text{App } t) \text{enum}_{\text{syn}}^{\sigma} (\text{Tree } =_{\beta\eta}) \text{fmap} (\text{quote}^{\tau} \circ f) \text{enum}_{\text{set}}^{\sigma}$$

For proof, we notice that, by Lemma 5.1 (1) below for σ ,

$$\text{enum}_{\text{syn}}^{\sigma} (\text{Tree } R^{\sigma}) \text{enum}_{\text{set}}^{\sigma}$$

Hence, by assumption and the fact that `fmap` commutes with the effect on relations

$$\text{fmap} (\text{App } t) \text{enum}_{\text{syn}}^{\sigma} (\text{Tree } R^{\tau}) \text{fmap } f \text{enum}_{\text{set}}^{\sigma}$$

Hence, by IH of the Lemma for τ ,

$$\text{fmap} (\text{App } t) \text{enum}_{\text{syn}}^{\sigma} (\text{Tree } =_{\beta\eta}) \text{fmap} (\text{quote}^{\tau} \circ f) \text{enum}_{\text{set}}^{\sigma}$$

□

The proof above used two lemmas. One is essentially free, but the other is technical.

Lemma 5.1 (“Free” Lemma).

1. $\text{enum}_{\text{syn}}^{\sigma} (\text{Tree } R^{\sigma}) \text{enum}_{\text{set}}^{\sigma}$.
2. $\text{questions}_{\text{syn}}^{\sigma} [R^{\sigma} \rightarrow R^{\text{Bool}}] \text{questions}_{\text{set}}^{\sigma}$.

Proof. The proof is simultaneous for (1) and (2) by induction on σ .

- Case `Bool`: Trivial.
- Case $\sigma \rightarrow \tau$: Proof of (1) uses IH (2) for σ and IH (1) for τ ; proof of (2) uses IH (1) for σ and IH (2) for τ .

□

Lemma 5.2 (Technical Lemma). For $t \in \text{Tm}_{\Gamma} \sigma$:

$$t =_{\beta\eta} \text{find}_{\text{syn}} [q \ t \mid q \leftarrow \text{questions}_{\text{syn}}^{\sigma}] \text{enum}_{\text{syn}}^{\sigma}$$

Proof. By induction on σ .

- Case `Bool`:

$$\begin{aligned} t &= \text{If } t \ \text{True} \ \text{False} \\ &= \text{If } t \ (\text{find}_{\text{syn}} [] (\text{Val True})) \ (\text{find}_{\text{syn}} [] (\text{Val False})) \\ &= \text{find}_{\text{syn}} [t] \ (\text{Choice} (\text{Val True}) (\text{Val False})) \\ &= \text{find}_{\text{syn}} [q \ t \mid q \leftarrow \text{questions}_{\text{syn}}^{\text{Bool}}] \ \text{enum}_{\text{syn}}^{\text{Bool}} \end{aligned}$$

– Case $\sigma : \rightarrow \tau$:

$$\begin{aligned}
t &=_{\beta\eta} \text{Lam}^\sigma z (\text{App } t (\text{Var } z)) \quad (z \text{ fresh wrt } \Gamma) \\
&=_{\beta\eta} \text{(by IH for } \sigma) \\
&\quad \text{Lam}^\sigma z (\text{find}_{\text{syn}} [q(\text{Var } z) \mid q \leftarrow \text{questions}_{\text{syn}}^\sigma] \text{enum}_{\text{syn}}^\sigma) \\
&=_{\beta\eta} \text{(by Sublemma below)} \\
&\quad \text{Lam}^\sigma z \\
&\quad (\text{find}_{\text{syn}} [q (\text{App } t u) \mid u \leftarrow \text{flatten } \text{enum}_{\text{syn}}^\sigma, q \leftarrow \text{questions}_{\text{syn}}^\tau] \\
&\quad (\text{fmap } (\lambda g g (\text{Var } z)) (\text{mkenum}_{\text{syn}} \text{questions}_{\text{syn}}^\sigma \text{enum}_{\text{syn}}^\tau))) \\
&=_{\beta\eta} \text{by Lemma 5.3 and functor laws} \\
&\quad \text{find}_{\text{syn}} [q (\text{App } t u) \mid u \leftarrow \text{flatten } \text{enum}_{\text{syn}}^\sigma, q \leftarrow \text{questions}_{\text{syn}}^\tau] \\
&\quad (\text{fmap } (\text{lam } \sigma) (\text{mkenum}_{\text{syn}} \text{questions}_{\text{syn}}^\sigma \text{enum}_{\text{syn}}^\tau)) \\
&= \text{find}_{\text{syn}} [q t \mid q \leftarrow \text{questions}_{\text{syn}}^{\sigma \rightarrow \tau}] \text{enum}_{\text{syn}}^{\sigma \rightarrow \tau}
\end{aligned}$$

The sublemma is: Given $qs \diamond us$ then

$$\begin{aligned}
&\text{App } t (\text{find}_{\text{syn}} [q u \mid q \leftarrow qs] us) \\
&=_{\beta\eta} \text{find}_{\text{syn}} [q (\text{App } t u') \mid u' \leftarrow \text{flatten } us, q \leftarrow \text{questions}_{\text{syn}}^\tau] \\
&\quad (\text{fmap } (\lambda g g u) (\text{mkenum}_{\text{syn}} qs \text{enum}_{\text{syn}}^\tau))
\end{aligned}$$

The proof is by induction on $qs \diamond us$.

- Case $[] \diamond \text{Val } u^*$:

Assume $u =_{\beta\eta} \text{find}_{\text{syn}} [] (\text{Val } u^*)$, i.e., $u =_{\beta\eta} u^*$. We get

$$\begin{aligned}
&\text{App } t (\text{find}_{\text{syn}} [q u \mid q \leftarrow qs] us) \\
&=_{\beta\eta} \text{(by IH of the Lemma for } \tau) \\
&\quad \text{find}_{\text{syn}} [q (\text{App } t u^*) \mid q \leftarrow \text{questions}_{\text{syn}}^\tau] \text{enum}_{\text{syn}}^\tau \\
&= \text{find}_{\text{syn}} [q (\text{App } t u^*) \mid q \leftarrow \text{questions}_{\text{syn}}^\tau] \\
&\quad (\text{fmap } (\lambda g g u) (\text{fmap } (\lambda v \lambda u' v) \text{enum}_{\text{syn}}^\tau)) \\
&= \text{find}_{\text{syn}} [q (\text{App } t u') \mid u' \leftarrow [u^*], q \leftarrow \text{questions}_{\text{syn}}^\tau] \\
&\quad (\text{fmap } (\lambda g g u) (\text{mkenum}_{\text{syn}} [] \text{enum}_{\text{syn}}^\tau))
\end{aligned}$$

- Case $q : qs \diamond \text{Choice } l r$:

$$\begin{aligned}
&\text{App } t (\text{find}_{\text{syn}} [q' u \mid q' \leftarrow q : qs] (\text{Choice } l r)) \\
&= \text{App } t (\text{if}' (q u) (\text{find}_{\text{syn}} [q' u \mid q' \leftarrow qs] l) \\
&\quad (\text{find}_{\text{syn}} [q' u \mid q' \leftarrow qs] r)) \\
&=_{\beta\eta} \text{if}' (q u) (\text{App } t (\text{find}_{\text{syn}} [q' u \mid q' \leftarrow qs] l)) \\
&\quad (\text{App } t (\text{find}_{\text{syn}} [q' u \mid q' \leftarrow qs] r)) \\
&=_{\beta\eta} \text{(by IH of the Sublemma for } qs \diamond l, qs \diamond r) \\
&\quad \text{if}' (q u) \\
&\quad (\text{find}_{\text{syn}} [q' (\text{App } t u') \mid u' \leftarrow \text{flatten } l, q' \leftarrow \text{questions}_{\text{syn}}^\tau] \\
&\quad (\text{fmap } (\lambda g g u) (\text{mkenum}_{\text{syn}} qs \text{enum}_{\text{syn}}^\tau)))
\end{aligned}$$

$$\begin{aligned}
& (\text{find}_{\text{syn}} [q' (\text{App } t \ u') \mid u' \leftarrow \text{flatten } r, q' \leftarrow \text{questions}_{\text{syn}}^{\tau}] \\
& \quad (\text{fmap } (\lambda g \ g \ u) (\text{mkenum}_{\text{syn}} \ q s \ \text{enum}_{\text{syn}}^{\tau}))) \\
= & \text{find}_{\text{syn}} [] (\text{Val } (\text{If}' (q \ u) \\
& \quad (\text{find}_{\text{syn}} [q' (\text{App } t \ u') \mid u' \leftarrow \text{flatten } l, q' \leftarrow \text{questions}_{\text{syn}}^{\tau}] \\
& \quad \quad (\text{fmap } (\lambda g \ g \ u) (\text{mkenum}_{\text{syn}} \ q s \ \text{enum}_{\text{syn}}^{\tau}))) \\
& \quad \quad (\text{find}_{\text{syn}} [q' (\text{App } t \ u') \mid u' \leftarrow \text{flatten } r, q' \leftarrow \text{questions}_{\text{syn}}^{\tau}] \\
& \quad \quad \quad (\text{fmap } (\lambda g \ g \ u) (\text{mkenum}_{\text{syn}} \ q s \ \text{enum}_{\text{syn}}^{\tau})))))) \\
=_{\beta\eta} & (\text{by twice Lemma 5.4}) \\
& \text{find}_{\text{syn}} ([q' (\text{App } t \ u') \mid u' \leftarrow \text{flatten } l, q' \leftarrow \text{questions}_{\text{syn}}^{\tau}] \\
& \quad ++ ([q' (\text{App } t \ u') \mid u' \leftarrow \text{flatten } r, q' \leftarrow \text{questions}_{\text{syn}}^{\tau}] \\
& \quad ++ [])) \\
& \quad ((\text{fmap } (\lambda g \ g \ u) (\text{mkenum}_{\text{syn}} \ q s \ \text{enum}_{\text{syn}}^{\tau})) \gg= \lambda v_0 \\
& \quad \quad (\text{fmap } (\lambda g \ g \ u) (\text{mkenum}_{\text{syn}} \ q s \ \text{enum}_{\text{syn}}^{\tau})) \gg= \lambda v_1 \\
& \quad \quad \text{Val } (\text{If}' (q \ u) \ v_0 \ v_1)) \\
= & \text{find}_{\text{syn}} ([q' (\text{App } t \ u') \mid u' \leftarrow \text{flatten } l, q' \leftarrow \text{questions}_{\text{syn}}^{\tau}] \\
& \quad ++ [q' (\text{App } t \ u') \mid u' \leftarrow \text{flatten } r, q' \leftarrow \text{questions}_{\text{syn}}^{\tau}]) \\
& \quad (\text{fmap } (\lambda g \ g \ u) \\
& \quad \quad ((\text{mkenum}_{\text{syn}} \ q s \ \text{enum}_{\text{syn}}^{\tau}) \gg= \lambda g_0 \\
& \quad \quad \quad (\text{mkenum}_{\text{syn}} \ q s \ \text{enum}_{\text{syn}}^{\tau}) \gg= \lambda g_1 \\
& \quad \quad \quad \text{Val } (\lambda u' \ \text{If}' (q \ u') (g_0 \ u') (g_1 \ u')))) \\
= & \text{find}_{\text{syn}} [q' (\text{App } t \ u') \mid t \leftarrow (\text{flatten } l) ++ (\text{flatten } r), \\
& \quad \quad \quad q' \leftarrow \text{questions}_{\text{syn}}^{\tau}] \\
& \quad (\text{fmap } (\lambda g \ g \ u) (\text{mkenum}_{\text{syn}} (q : q s) \ \text{enum}_{\text{syn}}^{\tau})) \\
= & \text{find}_{\text{syn}} [q' (\text{App } t \ u') \mid u' \leftarrow \text{flatten } (\text{Choice } l \ r), \\
& \quad \quad \quad q' \leftarrow \text{questions}_{\text{syn}}^{\tau}] \\
& \quad (\text{fmap } (\lambda g \ g \ u) (\text{mkenum}_{\text{syn}} (q : q s) \ \text{enum}_{\text{syn}}^{\tau}))
\end{aligned}$$

□

We have used two lemmas, which are easy to prove:

Lemma 5.3. *Given $as \diamond us$ it is true that*

$$\text{find}_{\text{syn}} \ as \ (\text{fmap } f \ us) =_{\beta\eta} \ f \ (\text{find}_{\text{syn}} \ as \ us)$$

Proof. Simple induction on $as \diamond us$.

Lemma 5.4. *Given $as \diamond ts$ and, for all $u \in \text{Tm}_{\Gamma} \ \sigma$, $bs \diamond h \ u$, then*

$$\text{find}_{\text{syn}} (as ++ bs) (ts \gg= h) =_{\beta\eta} \ \text{find}_{\text{syn}} \ bs \ (h \ (\text{find}_{\text{syn}} \ as \ ts))$$

Proof. By induction on $as \diamond ts$.

– Case $[] \diamond \text{Val } t$:

$$\begin{aligned}
& \text{find}_{\text{syn}} ([] ++ bs) ((\text{Val } t) \gg= h) \\
&= \text{find}_{\text{syn}} bs ((\text{Val } t) \gg= h) \\
&= \text{find}_{\text{syn}} bs (h t) \\
&= \text{find}_{\text{syn}} bs (h (\text{find}_{\text{syn}} [] (\text{Val } t)))
\end{aligned}$$

– Case $a : as \diamond \text{Choice } l r$:

$$\begin{aligned}
& \text{find}_{\text{syn}} ((a : as) ++ bs) ((\text{Choice } l r) \gg= h) \\
&= \text{find}_{\text{syn}} (a : (as ++ bs)) ((\text{Choice } l r) \gg= h) \\
&= \text{If}' a (\text{find}_{\text{syn}} (as ++ bs) (l \gg= h)) (\text{find}_{\text{syn}} (as ++ bs) (r \gg= h)) \\
&= (\text{by IH for } as \diamond l, as \diamond r) \\
&\quad \text{If}' a (\text{find}_{\text{syn}} bs (h (\text{find}_{\text{syn}} as l))) (\text{find}_{\text{syn}} bs (h (\text{find}_{\text{syn}} as r))) \\
&=_{\beta\eta} \text{find}_{\text{syn}} bs (h (\text{If}' a (\text{find}_{\text{syn}} as l) (\text{find}_{\text{syn}} as r))) \\
&= \text{find}_{\text{syn}} bs (h (\text{find}_{\text{syn}} (a : as) (\text{Choice } l r)))
\end{aligned}$$

□

6 Discussion and further work

Instead of decision trees we could have used a direct encoding of the graph of a function, we call this the truth-table semantics. However, this approach leads not only too much longer normal forms but also the semantic equality is less efficient. On the other hand it is possible to go further and use Binary Decision Diagrams (BDDs) [7] instead of decision trees. We plan to explore this in further work and also give a detailed analysis of the normal forms returned by our algorithm.

We have argued that λ^{-2} is the simplest λ -calculus with closed types, however we are confident that the technique described here works also for closed types in $\lambda^{0+1 \times \rightarrow}$ (the finitary λ -calculus). We leave this extension for a journal version of this work.

One can go even further and implement finitary Type Theory, i.e. $\lambda^{012\Sigma\Pi}$ (note that $A + B = \Sigma x \in 2. \text{if } x \text{ then } A \text{ else } B$). This could provide an interesting base for a type-theoretic hardware description and verification language.

The approach presented here works only for calculi without type variables. It remains open to see whether this approach can be merged with the the standard techniques for NBE for systems with type variables, leading to an alternative proof of completeness and maybe even finite completeness for the calculi discussed above.

References

1. T. Altenkirch, P. Dybjer, M. Hofmann, and P. Scott. Normalization by evaluation for typed lambda calculus with coproducts. In *Proc. of 16th Annual IEEE Symposium on Logic in Computer Science*, pages 303–310. IEEE CS Press, 2001.

2. T. Altenkirch, M. Hofmann, and T. Streicher. Categorical reconstruction of a reduction free normalization proof. In D. Pitt, D. E. Rydeheard, and P. Johnstone, editors, *Category Theory and Computer Science*, number 953 in LNCS, pages 182–199, Springer-Verlag, 1995.
3. T. Altenkirch, M. Hofmann, and T. Streicher. Reduction-free normalisation for a polymorphic system. In *Proc. of 11th Annual IEEE Symposium on Logic in Computer Science*, pages 98–106. IEEE CS Press, 1996.
4. T. Altenkirch, M. Hofmann, and T. Streicher. Reduction-free normalisation for system F . Unpublished, available on WWW at <http://www.cs.nott.ac.uk/~txa/publ/f97.pdf>, 1997.
5. V. Balat. *Une étude des sommes fortes : isomorphismes et formes normales*. PhD thesis, Université Denis Diderot, 2002.
6. U. Berger and H. Schwichtenberg. An inverse of the evaluation functional for typed λ -calculus. In *Proc. of 6th Annual IEEE Symposium on Logic in Computer Science*, pages 202–211. IEEE CS Press, 1991.
7. R. E. Bryant. Graph-based algorithms for boolean function manipulation. *IEEE Trans. on Computers*, 35(8):677–691, 1986.
8. D. J. Dougherty and R. Subrahmanyam. Equality between functionals in the presence of coproducts. *Information and Computation*, 157(1–2):52–83, 2000.
9. P. Dybjer and A. Filinski. Normalization and partial evaluation. In G. Barthe et al., editors, *Applied Semantics*, number 2395 in LNCS, pages 137–192. Springer-Verlag, 2002.
10. N. Ghani. *Adjoint Rewriting*. PhD thesis, LFCS, Univ. of Edinburgh, 1995.
11. N. Ghani. $\beta\eta$ -equality for coproducts. In M. Dezani-Ciancaglini and G. D. Plotkin, editors, *Typed Lambda Calculus and Applications*, number 902 in LNCS, pages 171–185. Springer-Verlag, 1995.
12. C. McBride and J. McKinna. The view from the left. To appear in the Journal of Functional Programming, Special Issue: Dependent Type Theory Meets Programming Practice, 2004.
13. G. D. Plotkin. Lambda-definability and logical relations. Memorandum SAI-RM-4, Lab. for Artif. Intell., Univ. of Edinburgh, Oct. 1973.